# LuckPay Web Manage System
#
# Copyright (c) 2016 Lucky Byte, Inc.
#
$ ->
    form = $('form.main')
    form.form
        inline : true,
        fields : name : 'empty'

    # 交易权限全选按钮事件
    form.find('.ui.button#trcode_all').click ->
        form.find('.checkbox.trcode').each ->
            $(this).checkbox 'check'

    # 交易权限全不选按钮事件
    form.find('.ui.button#trcode_none').click ->
        form.find('.checkbox.trcode').each ->
            $(this).checkbox 'uncheck'

    # 恢复交易权限值
    trctl = form.form 'get value', 'trctl'
    if trctl.trim().length > 0
        trctl = JSON.parse trctl

        # 交易权限
        form.find('.checkbox.trcode').checkbox 'uncheck'
        $.each trctl.trcodes, (k, v) ->
            form.find(".checkbox.trcode.#{k}").checkbox 'check'
        form.form 'set value', 'trctl', ''

        # 交易限制
        for trcode in [ 't20', 't60', 'tz0' ]
            continue if not trctl[trcode]
            for field in window.trctl_fields
                if trctl[trcode][field]
                    form.find(".fields.trctl.#{trcode} input[name=#{field}]")
                        .val(trctl[trcode][field])

    # 提交前处理，组织上传数据
    form.submit (evt) ->
        return evt.preventDefault() if not $(this).form('is valid')

        trctl = {}

        # 组织交易权限
        trctl.trcodes = {}
        form.find('.checkbox.trcode').each ->
            if $(this).checkbox 'is checked'
                trctl.trcodes[$(this).find('input').attr('name')] = true

        # 交易限制
        for trcode in [ 't20', 't60', 'tz0' ]
            trctl[trcode] = {}
            for field in window.trctl_fields
                val = form.find(".fields.trctl.#{trcode} input[name=#{field}]").val()
                trctl[trcode][field] = val if val and val.trim().length > 0

        form.form 'set value', 'trctl', JSON.stringify trctl
